
... Microsoft Excel

= Tabellenkalkulationsprogramm von Microsoft Office - Paket.
= Darstellung von Daten in Tabellenform.

= Kalkulation von Daten aus vorhanden Daten.

= Berechnung von Daten mit Hilfe von Formeln.

= Weiterverarbeitung in Diagrammen und PivotTabellen.

Modell ,Microsoft Excel” (Ausschnitt)

Application
(Excel)

Workbook
(Arbeitsmappe)

Worksheet
(Arbeitsblatt)

Range Cells
(Zellbereich) (Zelle)

... aus Microsoft Access aufrufen

= Voraussetzung: Excel ist auf dem Entwicklungsrechner
vorhanden.

= Zuerst muss ein Verweis auf das Objektmodell angelegt.
= Excel wird geoffnet.
= Arbeitsmappen werden neu angelegt oder geoffnet.

= Daten werden aus einem Tabellenblatt gelesen oder in dieses
geschrieben.

Einbindung des Objektmodells

= Extras — Verweise im VBA-Editor.

= Mit einem Klick auf das Kastchen wird die Bibliothek
Microsoft Excel 14.0 Object Library eingebunden. Die
Version bezieht sich auf Microsoft Excel 2010.

= Durch Aktivierung des Verweises wird das Objektmodell von
Excel bekannt.

Offnen von Microsoft Excel

Dim appExcel As Excel.Application

Set appExcel = New Excel. Application

Visual Basic for Application Seite 6



g% Leibniz Universitat
I 5 IT Services

Erlauterung

= Early Bindung.

= Voraussetzung: Der Verweis auf die Anwendung ,Microsoft
Excel” muss vorhanden sein.

= Die Objekt-Variable kann nur auf Objekte von der Klasse
Excel” verweisen. Mit Hilfe von New wird ein Verweis auf eine
nicht geoffnete Microsoft Excel — Anwendung ubergeben.

= Beim Kompilieren des VBA-Codes wird eine Syntaxprufung
durchgefuhrt.

Offnen von Microsoft Excel

Dim appExcel As Object

Set appExcel = CreateObject("Excel. Application")

Erlauterung

= Late Binding.
= Die Objekt-Variable kann jeden beliebigen Typ von Klasse
aufnehmen.

= Mit Hilfe von CreateObjekt() wird eine Instanz der
entsprechenden Klasse erzeugt. Der Funktion wird der Name
der gewunschten Klasse in runden Klammern ubergeben.

= Es werden keine Verweise benaotigt.

Einblenden der Anwendung

appExcel. Visible = True

= Die Methode .Visible blendet die Microsoft Office — Anwendung
ein (True) oder aus (False).

Schliel3en von Microsoft Excel

appExcel.Quit

= Die Methode .Quit schlief3t die Microsoft Office - Anwendung.

Arbeitsmappe (Workbook)

= Datei mit der Endung ,xIs” oder ,xIsx” ab Excel 2007.

= Container fur alle Arbeitsblatter eines Microsoft Excel -
Projekts.

= Die Arbeitsmappe enthalt standardmalig drei Arbeitsblatter.

Objekt-Variable vom Typ ,Workbook"

Dim workbook As Excel.workbook

= Die Objekt-Variable kann Verweise auf eine Arbeitsmappe
speichern.

Visual Basic for Application Seite 13



'r Leibniz Universitat
I 2 IT Services

Anlegen einer neuen Arbeitsmappe

Dim workbook As Excel.workbook

Set workbook = appExcel. Workbooks.Add

= Die Auflistung .Workbooks enthalt alle, in einer Microsoft Excel —

Anwendung geoffneten, Arbeitsmappen.
= Mit Hilfe der Methode .Add wird eine neue Arbeitsmappe

erzeugt.

Schlielden der Arbeitsmappe

Dim workbook As Excel.workbook

Set workbook = appExcel. Workbooks.Add
workbook.Close

= Mit Hilfe der Methode .Close wird eine Arbeitsmappe
geschlossen.

Arbeitsblatter (Worksheet)

Dim worksheet As Excel.worksheet

= Tabellenblatter in Excel bestehen aus Zeilen und Spalten.

Aktivierung eines Arbeitsblattes

Set worksheet = workbook. Worksheets(1)

= Die Auflistung Worksheets enthalt alle Arbeitsblatter in einer

Arbeitsmappe.
In den runden Klammern wird der Auflistung ein Index zur
|dentifizierung eines Arbeitsblattes Ubergeben. Die

Arbeitsblatter werden von 1 bis n nummeriert.

Name eines Arbeitsblattes

Set worksheet = workbook. Worksheets(1)
worksheet.Name = "Kunde"

= Das Attribut .Name gibt Auskunft Gber den Namen eines
Arbeitsblatts.

Speicherung eines Arbeitsblattes

worksheet.SaveAs "Kundeadresse.xlsx"

= Die Methode .SaveAs speichert ein Arbeitsblatt unter dem

angegebenen Namen. )
= Die Methode .Save speichert Anderungen in einem

vorhandenen Arbeitsblatt.

Kopieren einer Tabelle von Access nach Excel

Dim dbs As DAO.Database

Dim rs As DAO.Recordset

Dim workbook As Excel.workbook
Dim worksheet As Excel.worksheet

Set workbook = appExcel. Workbooks.Add
Set worksheet = workbook. Worksheets(1)

Set dbs = Application.CurrentDb
Set rs = dbs.OpenRecordset("Kunden", dbOpenTable)

If Not (rs.BOF And rs.EOF) Then
worksheet.Range("A1").CopyFromRecordset rs
End If

rs.Close

Erlauterung der if-Anweisung

Set rs = dbs.OpenRecordset("Kunden", dbOpenTable)
If Not (rs.BOF And rs.EOF) Then

End If

= Wenn die Marker BOF ,Beginn des Recordsets” und EOF ,Ende
des Recordsets” nicht gleich sind ...
= Wenn die Tabelle nicht leer ist ...

Kopieren der Datensatze

worksheet.Range("A1").CopyFromRecordset rs

= Dem Zellbereich .Range wird der Name einer Zelle in runden
Klammern ubergeben. In diesem Beispiel wird auf die Zelle in
der Spalte A in der ersten Zeile in dem angegeben Arbeitsblatt
verwiesen.

= Beginnend mit dieser Zelle werden die Daten aus dem
Recordset mit Hilfe der Methode .CopyFromRecordset kopiert.

Arbeitsschritte der Methode ,,CopyFromRecordset

= Kopiere die zu exportierenden Datensatze in die
Zwischenablage.

= Markiere die Zelle A1 auf dem ersten Arbeitsblatt in der
geoffneten Arbeitsmappe.

= Fuge die in der Zwischenablage vorhandenen Datensatze ab
der markierten Zelle in das aktive Arbeitsblatt ein.

Setzen von ,Kopfzeilen®

Spalte = 1

For Each element In rs.Fields
worksheet.Cells(1, spalte).Value = element.Name

spalte = spalte + 1
Next

= Fur jedes Feld in dem Recordset wird eine Spaltentberschrift

gesetzt.
= Die Auflistung .Cells enthalt jede Zelle in einem Arbeitsblatt. Die

Auflistung ist mehrdimensional.

Index einer Zelle

Spalte = 1

For Each element In rs.Fields
worksheet.Cells(1, spalte).Value = element.Name
spalte = spalte + 1

Next

= Die Auflistung .Cells enthalt jede Zelle in einem Arbeitsblatt.

= Eine Zelle wird durch die Zeile und die Spalte eindeutig
identifiziert.

= Die Zeile und die Spalte werden in den runden Klammern durch
ein Kommata getrennt.

Inhalt einer Zelle

Spalte = 1

For Each element In rs.Fields
worksheet.Cells(1, spalte).Value = element.Name
spalte = spalte + 1

Next

= Die Eigenschaft .Value speichert den Inhalt einer Zelle.

Formatierung einer Zelle

worksheet.Cells(zeile, spalte).NumberFormat = "#,##0.00 €"

= Die Eigenschaft .NumberFormat formatiert Zahlen in einer Zelle
entsprechend eines Musters.

= Das Muster enthalt Formatierungszeichen wie zum Beispiel das
Hash-Zeichen oder die Null. Das Muster kann aber auch
Literale wie das Euro-Zeichen enthalten.

Formatierungszeichen

worksheet.Cells(zeile, spalte).NumberFormat = "#,##0.00 €"

= An der Position des Hash-Zeichens kann eine Zahl vorhanden
sein, muss aber nicht.

= An der Position des Formatierungszeichen ,0“ muss eine Zahl
vorhanden sein, andernfalls wird 0 ausgegeben.

= Das Komma symbolisiert das landerspezifische Tausender-
Trennzeichen.

= Der Punkt symbolisiert das Dezimaltrennzeichen.

Attribute VB_Name = "ExportToExcel"

Option Compare Database

Option Explicit



Sub openExcel()

    Dim appExcel As Excel.Application



    Set appExcel = New Excel.Application

    appExcel.Visible = True



    appExcel.Quit

End Sub



Sub openObjekt()

    Dim appExcel As Object



    Set appExcel = CreateObject("Excel.Application")

    appExcel.Visible = True



    appExcel.Quit

End Sub



Sub newWorkbook()

    Dim appExcel As Object

    Dim workbook As Excel.workbook



    Set appExcel = CreateObject("Excel.Application")

    appExcel.Visible = True



    Set workbook = appExcel.Workbooks.Add



    'workbook.Close

    'appExcel.Quit

End Sub



Sub workWithWorksheets()

    Dim appExcel As Object

    Dim workbook As Excel.workbook

    Dim worksheet As Excel.worksheet



    Set appExcel = CreateObject("Excel.Application")

    appExcel.Visible = True



    Set workbook = appExcel.Workbooks.Add

    Set worksheet = workbook.Worksheets(1)

    worksheet.Name = "Kunde"

    

    'workbook.Close

    'appExcel.Quit

End Sub



Sub copyRecordsets()

    Dim dbs As DAO.Database

    Dim rs As DAO.Recordset

    

    Dim appExcel As Object

    Dim workbook As Excel.workbook

    Dim worksheet As Excel.worksheet



    Set appExcel = CreateObject("Excel.Application")

    appExcel.Visible = True



    Set workbook = appExcel.Workbooks.Add

    Set worksheet = workbook.Worksheets(1)

    worksheet.Name = "Kunde"

    

    Set dbs = Application.CurrentDb

    Set rs = dbs.OpenRecordset("Kunden", dbOpenTable)

    

    If Not (rs.BOF And rs.EOF) Then

        worksheet.Range("A1").CopyFromRecordset rs

    End If

    

    rs.Close

    Set rs = Nothing

    

    Set dbs = Nothing

    

    worksheet.SaveAs "Kundeadresse.xlsx"

    'workbook.Close

    Set workbook = Nothing

    Set worksheet = Nothing

    

    'appExcel.Quit

    Set appExcel = Nothing

End Sub



Sub fillExcelCells()

    Dim dbs As DAO.Database

    Dim rs As DAO.Recordset

    Dim sqlAnweisung As String

    Dim spalte As Integer

    Dim zeile As Integer

    Dim element As Variant

    Dim curBestellpreis As Currency

    

    Dim appExcel As Object

    Dim workbook As Excel.workbook

    Dim worksheet As Excel.worksheet



    Set appExcel = CreateObject("Excel.Application")

    appExcel.Visible = True



    Set workbook = appExcel.Workbooks.Add

    Set worksheet = workbook.Worksheets(1)

    worksheet.Name = "Bestellungen"

    

    Set dbs = Application.CurrentDb

    Set rs = dbs.OpenRecordset("qryBestellung", dbOpenDynaset)

    

'    sqlAnweisung = "SELECT Bestellungen.[Bestell-Nr]"

'    sqlAnweisung = sqlAnweisung & ", Bestellungen.[Kunden-Code]"

'    sqlAnweisung = sqlAnweisung & ", Artikel.Artikelname"

'    sqlAnweisung = sqlAnweisung & ", Bestelldetails.Anzahl"

'    sqlAnweisung = sqlAnweisung & ", Bestelldetails.Einzelpreis"

'    sqlAnweisung = sqlAnweisung & ", Bestelldetails.Rabatt"

'    sqlAnweisung = sqlAnweisung & " FROM Artikel"

'    sqlAnweisung = sqlAnweisung & " RIGHT JOIN (Bestellungen"

'    sqlAnweisung = sqlAnweisung & " LEFT JOIN Bestelldetails"

'    sqlAnweisung = sqlAnweisung & " ON Bestellungen.[Bestell-Nr] = Bestelldetails.[Bestell-Nr])"

'    sqlAnweisung = sqlAnweisung & " ON Artikel.ID = Bestelldetails.[Artikel-Nr]"

'    sqlAnweisung = sqlAnweisung & " ORDER BY Bestellungen.[Bestell-Nr];"

'    Set rs = dbs.OpenRecordset(sqlAnweisung, dbOpenDynaset)

    

    spalte = 1

    For Each element In rs.Fields

        worksheet.Cells(1, spalte).Value = element.Name

        spalte = spalte + 1

    Next

    

    worksheet.Cells(1, spalte).Value = "Bestellpreis"

    

    rs.MoveFirst

    zeile = 2

    

    Do While Not (rs.EOF)

        spalte = 1

        

        If (rs.Fields("Artikelname") <> "") Then

            For Each element In rs.Fields

                worksheet.Cells(zeile, spalte).Value = element.Value

                spalte = spalte + 1

            Next

            

            curBestellpreis = 0

            If Nz(rs.Fields("Anzahl").Value, 0) > 0 Then

                If Nz(rs.Fields("Einzelpreis").Value, 0) > 0 Then

                    curBestellpreis = (rs.Fields("Anzahl").Value * rs.Fields("Einzelpreis").Value)

                    curBestellpreis = curBestellpreis - rs.Fields("Rabatt").Value

                End If

            End If

            

            worksheet.Cells(zeile, spalte).NumberFormat = "#,##0.00  €"

            worksheet.Cells(zeile, spalte).Value = curBestellpreis

            zeile = zeile + 1

            

        End If

        rs.MoveNext

    Loop

    

    rs.Close

    Set rs = Nothing

    

    Set dbs = Nothing

    

    worksheet.SaveAs Application.CurrentProject.Path & "\" & "Bestellungen.xlsx"

    'workbook.Close

    Set workbook = Nothing

    Set worksheet = Nothing

    

    'appExcel.Quit

    Set appExcel = Nothing

End Sub



